AD-A089 575 MARK RESOURCES INC MARINA DEL REY CA F/G 20/14
FORTRAN PROGRAM FOR HELICOPTER BLADE MODULATION SIGNAL.(U)
MAY 78 R L MITCHELL DAAK40-78-C-0031
UNCLASSIFIED MRI-149-19 NL

1 OF 1
AD A
089575

END
DATE
FILMED
10-80
DTIC

US ARMY MIRADCOM

FORTRAN PROGRAM FOR HELICOPTER BLADE MODULATION SIGNAL

TECH NOTE 105-042

22 MAY 78

PREPARED FOR: RF SYSTEMS BRANCH (DRDMI-TDR)
SYSTEMS SIMULATION DIRECTORATE
TECHNOLOGY LABORATORY
US ARMY MIRADCOM
REDSTONE ARSENAL, AL 35809

DAAK40-78-C-0031

PREPARED BY: DR. R. L. MITCHELL
MARK RESOURCES, INC.
4676 ADMIRALTY WAY
SUITE 303
MARINA DEL REY, CA 90291

MARK RESOURCES INCORPORATED

FORTRAN PROGRAM FOR HELICOPTER BLADE MODULATION SIGNAL

R. L. Mitchell

MRI Report 149-19
22 May 1978

The modulation signal for the rotating helicopter blade is derived in Reference 1. Two basic models were proposed in this reference: a model based on scattering from the blade tip and a specular flash model. Evidence presented in Reference 2 is somewhat inconclusive, although the statement is made that "most of the echo is derived from the outer portions of the rotor blade (approximately the outer 20% of the blade length)". Such a statement supports the model where scattering is assumed to occur at the blade tip.

In order to accommodate both of the above cases, a slightly more general model for the modulation signal of a set of rotating helicopter blades than given in References 3 and 4 is defined as

$$V(t) = \sum_{n=0}^{N-1} A(\theta_n)\,\mathrm{sinc}[2(L_{ep}/\lambda)\sin\theta_n]\,e^{j4\pi(L_{cp}/\lambda)\sin\theta_n} \qquad (1)$$

where

$$\mathrm{sinc}(x) = (\sin\pi x)/\pi x$$

$$\theta_n = 2\pi(f_s t + n/N) \qquad (2)$$

f_s = spin frequency

N = number of blades

$L_{ep} = L_e\cos\alpha$, L_e = effective blade length

$L_{cp} = L_c\cos\alpha$, L_c = location of phase center

α = angle between LOS and plane of rotation

λ = wavelength

The quantity $A(\theta_n)$ is used to allow different scattering properties on the leading and lagging edges of the blade. Since θ_n is referenced to the time of occurrence of the maximum Doppler, we can define

$$A(\theta_n) = \begin{cases} 1 \text{ for leading edge } (\cos\theta_n \geq 0) \\ \\ \rho \text{ for lagging edge } (\cos\theta_n < 0) \end{cases} \quad (3)$$

The instantaneous phase and frequency of the phase center of the n^{th} blade are given by

$$\phi_i = 4\pi(L_{cp}/\lambda)\sin\theta_n \quad (4)$$

$$f_i = 4\pi(L_{cp}/\lambda)f_s\cos\theta_n \quad (5)$$

where θ_n is given by (2).

From (5) we note that the maximum Doppler is given for $\cos\theta_n = 1$ as

$$f_{max} = 4\pi(L_{cp}/\lambda)f_s \quad (6)$$

In a typical case we might have $L_{cp} = 4$ m, $\lambda = .02$ m, and $f_s = 10$ Hz, so that $f_{max} = 25.1$ kHz. One half of a cycle later the minimum Doppler would be $-f_{max}$, so the bandwidth of the modulation signal is $2f_{max}$ or 50.2 kHz for the example. In order to use (1) to simulate the modulation signal, the sampling frequency must be at least as large as $2f_{max}$ to prevent aliasing or foldover (the amplitude modulation in the sinc function in (1) causes some additional Doppler broadening beyond $\pm f_{max}$). However, in the proposed application this would stipulate a sampling frequency of at least 50 kHz, which is about a factor of 4 larger than the receiver processing bandwidth. In other words, (1) simulates signals that would be outside of the receiver band, as well as the desired in-band signal.

In order to minimize the computation load it would be desirable to sample at the rate that is coincident with the receiver processing band. But in order to prevent the folding over of frequencies outside of the desired band, we must suppress the unwanted signals. This will be accomplished by computing (5), the instantaneous frequency of each component, to determine if that component is within the receiver processing band. If it is, then that component will be included in the summation in (1).

Disregarding the return from the hub portion of the blade assembly for the moment, it can be shown that as long as the number of blades is relatively small (≤ 6), the maximum number of components of (1) that can fall within a band that is 25% of $2f_{max}$ is two. Thus for N-2 of the components, only f_i in (5) need be computed. This result significantly reduces the computation time for $N > 2$.

The hub portion of the blade assembly also scatters energy. We can simulate this component of the modulation signal by the use of (1) if we redefine L_{ep} and L_{cp}. In general, the scattering will be much more isotropic than for the blade so that L_{ep} will be small (if $L_{ep} = 0$ then it would be isotropic). If the spectral band for the hub signal falls within the desired processing band, then occasionally the blade Doppler will also fall within this band. But this occurs only when the blade is near the quadrature points to the maximum and minimum Doppler. In other words, we will be looking at the ends of the blade. In order to conserve computer resources, the assumption is made that when the hub return falls within the processing band, the blade return will be neglected.

References

1. Mitchell, R. L., and I. P. Bottlik, "Techniques for Simulating Realistic RF Environment Signals on the RFSS," MRI Report 131-25, 28 February 1977.
2. "Improved Hawk Systems Threat Definition and Capabilities Against Helicopter and Liaison Aircraft," Raytheon Final Report BR-6501, Revision B, 15 Jan 1975.
3. Mitchell, R. L., "Helicopter Blade Modulation Model (Revised)," MRI Report 149-16, 29 March 1978.
4. McPherson, D. A., "Rotating Blade Modulation Waveform," Boeing/Huntsville Report, 15 March 1978.

Appendix

Program Listings

Routine	Comments
TEST	Test program
HBMSD	Set-up parameters for call to HBMS. Probably will be done in Datacraft/6.
HBMS	Generates modulation signal. Probably will be done in AP120B.
SINCOS	Fortran version of subroutine to be implemented as table lookup.

```
      PROGRAM TEST(OUTPUT)
      COMMON /T1/ VR(200)
      COMMON /T2/ VI(200)
      COMMON /D1/ NSAMP, FR, FS, F0, F1, R, DELH, WL
      COMMON /D2/ XFS, XF1, XF2, ASK, ABL, NBL, XLEP, XLCP, RHO, PEFF
      NSAMP=200
      FR=10000.
      FS=10.
      F0=0.
      F1=-5000.
      R=1000.
      DELH=0.
      WL=.02
      CALL HBMSD
      PRINT 100, (K, VR(K), VI(K), K=1, NSAMP)
      PRINT 101, NBL, XFS, XF1, XF2, ASK, ABL, XLEP, XLCP, RHO, PEFF
  100 FORMAT(I6, 2F12.6)
  101 FORMAT(/I6, 8F12.6, E16.6)
      END
```

```
      SUBROUTINE HBMSD
C
C DRIVER FOR HBMS
C
C IN THIS SUBROUTINE WE SET UP THE PARAMETERS FOR ONE CALL TO SUBROUTINE
C HBMS, WHICH GENERATES THE HELICOPTER BLADE MODULATION SIGNAL.
C
C THE REAL-TIME INPUTS THROUGH COMMON /D1/ ARE.....
C
C    NSAMP = NUMBER OF SAMPLES COMPUTED PER CALL TO HBMS
C       FR = SAMPLING FREQUENCY
C       FS = BLADE SPIN FREQUENCY
C       FO = DOPPLER FREQUENCY OF SKIN RETURN
C       F1 = LOWER FREQUENCY OF RECEIVER PROCESSING WINDOW
C        R = RANGE
C     DELH = DIFFERENCE IN ALTITUDE BETWEEN RADAR AND HELICOPTER
C       WL = WAVELENGTH
C
C THE OUTPUTS THROUGH COMMON /D2/ ARE.....
C
C      XFS = FS/FR
C      XF1 = (F1-FO)/FR
C      XF2 = XF1+1.
C      ASK = REFERENCE AMPLITUDE OF SKIN RETURN
C      ABL = REFERENCE AMPLITUDE OF BLADE OR HUB RETURN
C      NBL = NUMBER OF BLADES
C     XLEP = NORMALIZED PROJECTED EFFECTIVE BLADE LENGTH
C     XLCP = NORMALIZED PROJECTED EFFECTIVE LOCATION OF PHASE CENTER
C      RHO = RATIO OF BLADE AMPLITUDE ON LEADING EDGE TO LAGGING EDGE
C     PEFF = EFFECTIVE POWER TO BE RADIATED FROM RFSS ARRAY
C
C ALL OF THE REMAINING PARAMETERS ARE NON-REAL-TIME AND ARE SET IN THIS
C SUBROUTINE.  FIRST WE HAVE.....
C
C       AS = AMPLITUDE OF SKIN RETURN (=SQRT(RCS), NONFLUCTUATING)
C   PTGDSQ = PRODUCT OF TRANSMIT POWER, GAIN, SQUARE OF CHAMBER LENGTH
C
C THE BLADE PARAMETERS ARE.....
C
C     NBLB = NUMBER OF BLADES
C     ALEB = EFFECTIVE BLADE LENGTH
C     ALCB = EFFECTIVE LOCATION OF PHASE CENTER
C     RHCB = RATIO OF LEADING TO LAGGING EDGE AMPLITUDE FOR BLADE
C       AB = PEAK AMPLITUDE OF BLADE RETURN (=SQRT(RCS))
C
C THE HUB PARAMETERS ARE.....
C
C     NBLH = NUMBER OF HUB SECTIONS (USUALLY EQUAL TO NBLB)
C     ALEH = EFFECTIVE BLADE LENGTH FOR HUB (PROBABLY SMALL)
C     ALCH = EFFECTIVE LOCATION OF PHASE CENTER FOR HUB
```

```
C      RHOH = RATIO OF LEADING TO LAGGING EDGE AMPLITUDE FOR HUB
C        AH = PEAK AMPLITUDE OF HUB RETURN (=SQRT(RCS))
C
C THE TARGET IS ASSUMED TO BE NONFLUCTUATING.
C
C THE UNIT OF DISTANCE MUST BE CONSISTENT THROUGHOUT.
C
      COMMON /D1/ NSAMP,FR,FS,FO,F1,R,DELH,WL
      COMMON /D2/ XFS,XF1,XF2,ASK,ABL,NBL,XLEP,XLCP,RHO,PEFF

      DATA NBLH,ALEH,ALCH,RHOH,AH/3,0.,.5,1.,1./
      DATA NBLB,ALEB,ALCB,RHOB,AB/3,2.,8.,.3,1./
      DATA AS,PTGDSQ/1.,1./
      DATA FOURPI/12.566370614/
C
C COMPUTE COSINE OF ASPECT ANGLE
C
      CA=SQRT(1.-(DELH/R)**2)
C
C COMPUTE NORMALIZED FREQUENCIES
C
      XFS=FS/FR
      XFO=FO/FR
      XF1=F1/FR-XFO
      XF2=XF1+1.
C
C DETERMINE IF HUB DOPPLER IS VISIBLE
C
      IF(XF1.LT.O..AND.XF2.GT.O.) GO TO 20
C
C COMPUTE BLADE PARAMETERS
C
      NBL=NBLB
      XLEP=ALEB*CA/WL
      XLCP=ALCB*CA/WL
      RHO=RHOB
      APEAK=AS+2.*AB
      ASK=AS/APEAK
      ABL=AB/APEAK
      GO TO 30
C
C COMPUTE HUB PARAMETERS
C
   20 NBL=NBLH
      XLEP=ALEH*CA/WL
      XLCP=ALCH*CA/WL
      RHO=RHOH
      APEAK=AS+2.*AH
      ASK=AS/APEAK
      ABL=AH/APEAK
```

```
30 PEFF=PTGDSQ*(APEAK**2)/(FOURPI*R**4)
   CALL HBMS
   RETURN
   END
```

```
      SUBROUTINE HBMS
C
C HELICOPTER BLADE MODULATION SIGNAL
C
C IN THIS SUBROUTINE WE COMPUTE NSAMP COMPLEX SAMPLES IN THE ARRAY-PAIR
C (VR,VI) OF THE HELICOPTER BLADE MODULATION SIGNAL AT THE SAMPLE RATE
C WHICH WILL BE DESIGNATED AS FR.  ALL INPUT FREQUENCIES ARE NORMALIZED
C TO THIS QUANTITY, NAMELY
C
C         XFS = FS/FR    (FS = BLADE SPIN FREQUENCY)
C         XF1 = F1/FR    (F1 = LOWER FREQUENCY OF PROCESSING WINDOW)
C         XF2 = F2/FR    (F2 = UPPER FREQUENCY OF PROCESSING WINDOW)
C
C F1 AND F2 ARE REFERENCED TO THE SKIN DOPPLER (DC).
C
C WE ASSUME THAT THE PROCESS IS UNDERSAMPLED SO THAT ONLY THOSE DOPPLER
C FREQUENCIES BETWEEN F1 AND F2 WILL BE SIMULATED (IF ALL FREQUENCIES
C WERE TO BE SIMULATED THEN FOLDOVER WOULD RESULT WHENEVER 8*PI*XLCP*FS
C EXCEEDED F2-F1).  IN ADDITION, WE HAVE ON INPUT
C
C        XLEP = LEP/WL (LEP=EFFECTIVE PROJECTED BLADE LENGTH)
C        XLCP = LCP/WL (LCP=EFFECTIVE PROJECTED LOCATION OF PHASE CENTER)
C         ASK = REFERENCE AMPLITUDE FOR SKIN RETURN
C         ABL = REFERENCE AMPLITUDE FOR BLADE RETURN
C         RHO = RATIO OF BLADE AMPLITUDE ON LEADING EDGE TO LAGGING EDGE
C         NBL = NUMBER OF BLADES
C       NSAMP = NUMBER OF SAMPLES SIMULATED PER CALL TO HBMS
C
C THE BLADE LENGTH REFERS TO THE RADIUS OF THE ORBIT.
C
C IN THE USE OF THIS SUBROUTINE IT IS ASSUMED THAT XLEP AND XLCP WILL
C DESIGNATE THE PARAMETERS OF THE HUB PORTION OF THE BLADE ASSEMBLY IF
C F1 AND F2 ENCOMPASS DC (OR THE DOPPLER OF THE SKIN RETURN).  OTHERWISE
C THE PARAMETERS WILL BE FOR THE ACTUAL BLADE.
C
      COMMON /D1/ NSAMP
      COMMON /D2/ XFS,XF1,XF2,ASK,ABL,NBL,XLEP,XLCP,RHO,PEFF
      COMMON /T1/ VR(1)
      COMMON /T2/ VI(1)
      DATA TWOPI/6.283185307/
      DATA PHS/0./
      XN=1./FLOAT(NBL)
      XLCP2=XLCP*2.
      X1=TWOPI*XLCP2*XFS
      X2=TWOPI*XLEP
      DO 30 I=1,NSAMP
      UR=ASK
      UI=0.
      DO 20 L=1,NBL
      CALL SINCOS(S1,C1,PHS)
```

```
      XFI=X1*C1
      IF(XFI.LT.XF1) GO TO 15
      IF(XFI.GT.XF2) GO TO 15
      ARG=XLCP2*S1
      CALL SINCOS(S2,C2,ARG)
      ARG=X2*S1
      A=ABL
      IF(ABS(ARG).GT..001) A=ABL*SIN(ARG)/ARG
      IF(C1.LT.0.) A=RHO*A
      UR=UR+A*C2
      UI=UI+A*S2
   15 PHS=PHS+XN
   20 CONTINUE
      VR(I)=UR
      VI(I)=UI
      PHS=PHS+XFS
   30 CONTINUE
      PHS=PHS-FLOAT(NSAMP)
      RETURN
      END
```

```
SUBROUTINE SINCOS(S,C,X)
DATA TWOPI/6.283185307/
S=SIN(TWOPI*X)
C=COS(TWOPI*X)
RETURN
END
```